Includes barcode for easy software entry.

HEWLETT-PACKARD

HP-41

USERS' LIBRARY SOLUTIONS

Geometry



NOTICE

The program material contained herein is supplied without representation or warranty of any kind. Hewlett-Packard Company therefore assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof.

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press [■] [GTO] [.] [.] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⁜. When you see ⁜ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press [■] [APPEND] in ALPHA mode (press [■] and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: [■], followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [.] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: [■] [.] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing [■] and then the indirect address. Stack addresses are specified by pressing [.] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing [■] [.] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01◆LBL "SAMPLE"	[■] [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL"SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02"THIS IS A*
03 "⊢SAMPLE"	[ALPHA] [■] [APPEND] SAMPLE	*03" ⊢ SAMPLE*
04 AVIEW	[■] AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↗*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] [■] [.] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= [■] [ARCL] 03	*11"R3=*
12 ARCL 03	[■] [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	[■] [RTN]	*14 RTN*

TABLE OF CONTENTS

SINE PLATE SOLUTIONS, COORDINATE OF A POINT, POSITION AND SLOPE OF AN INCLINED HOLE

This program, with the aid of commonly available dowel pins and measuring tools, (and in the case of the sine plate, a sine plate and height blocks), will aid in accurately finding angles and heights for sine plates, position and slope of inclined holes and coordinates of points. All angular output is in decimal degrees.

Solution for Finding Coordinates of a Point:

Given: a, b, d and e, determine x and y

$$c = d^2 + e^2$$

$$\cos \alpha = \frac{b^2 + c^2 - a^2}{2bc}$$

$$\tan \theta = \frac{e}{d}$$

$$x = b \cos (\alpha + \theta)$$

$$y = b \sin (\alpha + \theta)$$



Solution for Finding the Location and Angle of an Inclined Hole:

Given: a, b, r_1, and d_2, determine θ and x

$$\sin \theta = \frac{2r_1 + d_2}{a_1 - b_1}$$

$$x = \tfrac{1}{2}(a_1 + b_1) + r_1 \cot \theta$$



Interchangeable Solutions for Work with a Sine Bar:

Given: e and k, determine α

$$\sin \alpha = \frac{e}{k}$$



Reference: HP-67/97 "Geometry" Users' Library Solutions Book.

Example:

Given: $a = 1.630''$ $r_1 = .200''$

$b = .260''$ $d_2 = .4375''$

Find θ, x of an inclined hole.

Keystrokes:	Display:
[shift] [fix] 4	(set display mode)
[XEQ] [ALPHA] SIZE [ALPHA] 003	
[XEQ] [ALPHA] SINP [ALPHA]	SINE PLATE
[B]	R1 ?
.2 [R/S]	d2 ?
.4375 [R/S]	a?
1.63 [R/S]	b?
.26 [R/S]	THETA=59.7007
[R/S]	X=1.0619

User Instructions

SIZE: 003

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set display mode			
2	Initialize the program.		[XEQ] SINP	SINE PLATE
3	To solve for the coordinates of a point:		[A]	a?
	Input a	a	[R/S]	b?
	b	b	[R/S]	d?
	d	d	[R/S]	e?
	and e.	e	[R/S]	X=(x)
			[R/S]	Y=(y)
4	To solve for the location and angle of an			
	inclined hole.		[B]	R1 ?
	Input r_1	r_1	[R/S]	d2 ?
	d_2	d_2	[R/S]	a?
	a	a	[R/S]	b?
	and b.	b	[R/S]	THETA=(θ)
			[R/S]	X=(x)
5	To solve for angles with a sine bar:		[C]	e?
	Input e	e	[R/S]	K?
	and K.	K	[R/S]	ALPHA=(α)
6	To solve for heights (of blocks) with a			
	sine bar:		[D]	K?
	Input K	K	[R/S]	ALPHA?
	and α.	α	[R/S]	e=(e)

Program Listings

```
 01◆LBL "SIN
P"
 02 SF 21
 03 SF 27
 04 DEG
 05 "SINE PL
ATE"
 06 AVIEW
 07◆LBL A
 08 "a?"
 09 PROMPT
 10 X↑2
 11 "b?"
 12 PROMPT
 13 STO 00
 14 X↑2
 15 -
 16 "d?"
 17 PROMPT
 18 "e?"
 19 PROMPT
 20 X<>Y
 21 R-P
 22 STO 01
 23 X<>Y
 24 STO 02
 25 RDN
 26 X↑2
 27 -
 28 CHS
 29 RCL 00
 30 /
 31 RCL 01
 32 /
 33 2
 34 /
 35 ACOS
 36 RCL 02
 37 +
 38 RCL 00
 39 P-R
 40 "X"
 41 XEQ 11
 42 RDN
 43 "Y"
 44 GTO 11
 45◆LBL B
 46 "R1 ?"
 47 PROMPT
 48 STO 00
 49 2
```

Initialize

Input a, b, d, and e

Calculate x,y

Input r_1, d_2, a, b

Calculate θ, x

```
 50 *
 51 "d2 ?"
 52 PROMPT
 53 +
 54 "a?"
 55 PROMPT
 56 RCL 00
 57 -
 58 STO 01
 59 "b?"
 60 PROMPT
 61 RCL 00
 62 +
 63 ST+ 01
 64 -
 65 /
 66 ASIN
 67 "THETA"
 68 GTO 11
 69 TAN
 70 1/X
 71 RCL 00
 72 *
 73 RCL 01
 74 2
 75 /
 76 +
 77 "X"
 78 XEQ 11
 79◆LBL C
 80 "e?"
 81 PROMPT
 82 "K?"
 83 PROMPT
 84 /
 85 ASIN
 86 "ALPHA"
 87 GTO 11
 88◆LBL D
 89 "K?"
 90 PROMPT
 91 "ALPHA?"
 92 PROMPT
 93 SIN
 94 *
 95 "e"
 96◆LBL 11
 97 "⊢="
 98 ARCL X
 99 AVIEW
100 END
```

Input e, k
Calculate α

Input α, k
Calculate e

Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	b or r_1	50	
	c or a_1, $a_1 + b_1$		
	θ		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 003 TOT. REG. 29 USER MODE
ENG ____ FIX ____ SCI ____ ON X OFF ____
DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21	S	Printer enable	Printer disable
27	S	User mode on	User mode off

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
Solve for coord.	A	Sine bar angles	C
Inclined hole	B	Sine bar height	D

SINE PLATE SOLUTIONS

PROGRAM REGISTERS NEEDED: 26

HEWLETT PACKARD
SOLUTIONS BOOK:
GEOMETRY

ROW 1 (1 : 4)

ROW 2 (5 : 7)

ROW 3 (7 : 15)

ROW 4 (16 : 24)

ROW 5 (25 : 37)

ROW 6 (38 : 45)

ROW 7 (45 : 51)

ROW 8 (51 : 59)

ROW 9 (59 : 67)

ROW 10 (67 : 77)

ROW 11 (77 : 82)

ROW 12 (83 : 88)

ROW 13 (89 : 93)

ROW 14 (94 : 100)

V NOTCHES AND LONG RADII

This program, together with commonly available dowel pins and height gages, will accurately determine the position and angles of "V" grooves or notches. With the same tools, long radii are accurately measured. All angular output is in decimal degrees.

Given: a, b, c, d, r_1 and r_2, determine x, y, α and β:

$$\tan \phi = \frac{b_1 - a_1}{d_1 - c_1}$$

$$\overline{O_1O_2} = \frac{d_1 - c_1}{\cos \phi}$$

$$\sin \theta = \frac{r_2 - r_1}{\overline{O_1O_2}}$$

$$\overline{OO_1} = \frac{r_1}{\sin \theta}$$

$$x = a_1 - \overline{OO_1} \sin \phi$$

$$y = c_1 - \overline{OO_1} \cos \phi$$

$$\alpha = 90° + \phi - \theta$$

$$\beta = 90° - \phi - \theta$$



Given: L, r and h, determine R:

$$(R + r)^2 = (R + h - r)^2 + (\tfrac{1}{2})\, L^2$$

$$R = \frac{L^2}{8(2r - h)} - \frac{h}{2}$$



Given L, r and h , determine R:

$$(R - r)^2 = (R - h_1 + r)^2 + (\tfrac{1}{2}\, L)^2$$

$$R = \frac{L^2}{8(h_1 - 2r)} + \frac{h_1}{2}$$



Reference: HP-67/97 "Geometry" Users' Library Solutions Book.

Example 1:

For Long Radius (concave arc)

L = 1.000''
r = .15625''
h = .270''

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 006	
[XEQ] [ALPHA] VNOTCH [ALPHA]	V NOTCHES, L.R.
[B]	L?
1 [R/S]	R?
.15625 [R/S]	H?
.27 [R/S]	R=2.8062

Example 2:

For "V" Notch

a = 1.500'' d = 2.800''
b = 2.125'' r_1 = .4375''
c = 1.750'' r_2 = .875''

Keystrokes:	Display:
[A]	R2?
.875 [R/S]	R1?
.4375 [R/S]	a?
1.5 [R/S]	b?
2.125 [R/S]	c?
1.75 [R/S]	d?
2.8 [R/S]	X=0.8750
[R/S]	Y=0.7000
[R/S]	ALPHA=63.9420
[R/S]	BETA=29.9010

User Instructions

SIZE: 006

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] VNOTCH	V NOTCHES, L.R.
3	To solve for "V" Notch:		[A]	R2?
	Input r_2	r_2	[R/S]	R1?
	r_1	r_1	[R/S]	a?
	a	a	[R/S]	b?
	b	b	[R/S]	c?
	c	c	[R/S]	d?
	and d and calculate X	d	[R/S]	X=(x)
	y		[R/S]	Y=(y)
	α		[R/S]	ALPHA=(α)
	and β.		[R/S]	BETA=(β)
4	To solve for long radii:			
4a	(concave radius)		[C]	L?
	Input L	L	[R/S]	R?
	r	r	[R/S]	H?
	h and calculate R.	h	[R/S]	R=(R)
4b	(convex radius)		[B]	L?
	Input L	L	[R/S]	R?
	r	r	[R/S]	H?
	h and calculate R	h	[R/S]	R=(R)

Program Listings

```
 01*LBL "VNO
TCH"
 02 CF 00
 03 "V NOTCH
ES, L.R."
 04 AVIEW
 05 STOP
 06*LBL C
 07 SF 00
 08*LBL B
 09 "L?"
 10 PROMPT
 11 X↑2
 12 "R?"
 13 PROMPT
 14 2
 15 *
 16 FS? 00
 17 CHS
 18 "H?"
 19 PROMPT
 20 FS?C 00
 21 CHS
 22 STO 00
 23 -
 24 8
 25 *
 26 /
 27 RCL 00
 28 2
 29 /
 30 -
 31 "R"
 32*LBL 11
 33 "⊢="
 34 ARCL X
 35 AVIEW
 36 STOP
 37 RTN
 38*LBL A
 39 "R2?"
 40 PROMPT
 41 STO 00
 42 "R1?"
 43 PROMPT
 44 STO 01
 45 -
 46 "a?"
 47 PROMPT
 48 LASTX
 49 -
 50 STO 02
 51 "b?"
 52 PROMPT
 53 RCL 00
 54 -
 55 -
 56 "c?"
 57 PROMPT
 58 RCL 01
 59 -
 60 STO 03
 61 "d?"
 62 PROMPT
 63 RCL 00
 64 -
 65 -
 66 STO 05
 67 /
 68 ATAN
 69 STO 04
 70 CLX
 71 RCL 05
 72 CHS
 73 RCL 04
 74 COS
 75 /
 76 RCL 00
 77 RCL 01
 78 -
 79 /
 80 1/X
 81 ASIN
 82 STO 05
 83 RCL 04
 84 RCL 01
 85 LASTX
 86 /
 87 P-R
 88 ST- 03
 89 RDN
 90 ST- 02
 91 RCL 04
 92 90
 93 RCL 05
 94 -
 95 +
 96 LASTX
 97 RCL 04
 98 -
 99 "X"
100 RCL 02
```

Initialize

Concave arcs

Input L, r, h

Calculate R

Display routine

Input a, b, c, d, r_1, and r_2

Calculate x, y, α, β

Display results

Program Listings

```
101 XEQ 11
102 "Y"
103 RCL 03
104 XEQ 11
105 "ALPHA"
106 RCL T
107 XEQ 11
108 "BETA"
109 RCL T
110 XEQ 11
111 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	r_2 or ± h	50	
	r_1		
	a_1, x		
	c_1, y		
	ϕ		
05	$c_1 - d_1$, θ	55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 006 TOT. REG. 33 USER MODE
ENG ____ FIX ____ SCI ____ ON X OFF ____
DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Concave arc	Convex arc

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

V NOTCHES AND LONG RADII

PROGRAM REGISTERS NEEDED: 28

HEWLETT PACKARD
SOLUTION BOOK:
GEOMETRY

ROW 1 (1 - 3)

ROW 2 (3 - 3)

ROW 3 (3 - 9)

ROW 4 (10 - 18)

ROW 5 (18 - 29)

ROW 6 (30 - 38)

ROW 7 (38 - 44)

ROW 8 (45 - 53)

ROW 9 (54 - 62)

ROW 10 (63 - 75)

ROW 11 (76 - 88)

ROW 12 (88 - 98)

ROW 13 (99 - 105)

ROW 14 (105 - 108)

ROW 15 (108 - 111)

INTERNAL AND EXTERNAL TAPERS

This program, used with commonly available dowel pins, height bases, and balls, will accurately determine the position and angle of both external and internal tapers. All angular output is in decimal degrees.

Internal Taper:

Given b, c, d, r_1 and r_2, determine C, D, ϕ and R_1

$$C^2 = 2c(r_1 + r_2) - c^2$$

$$D^2 = 2d(r_1 + r_2) - d^2$$

$$\tan \phi = \frac{D - C}{b}$$

$$2\theta = 90° + \phi$$

$$R_1 = C + r_1 \cot \theta$$



External Taper Case #1:

Given: a, b, r_1 and r_2, determine x and ϕ

$$\tan \theta = \frac{r_2 - r_1}{b_1 - a_1}$$

$$\phi = 90° - 2\theta$$

$$x = a_1 - r_1 \cot \theta$$



External Taper Case #2:

Given a, b, r and h, determine x and ϕ

$$\tan 2\theta = \frac{h}{b - a}$$

$$\phi = 90 - 2\theta$$

$$x = a - r - r\cot\theta$$



External Taper Case #3:

Given c, d, r_1 and r_2, determine x and ϕ

$$\sin\phi = \frac{r_2 - r_1}{d_1 - c_1}$$

$$x = \frac{r_1}{\cos\phi} - c_1 \tan\phi$$



Reference: HP-67/97 "Geometry" Users' Library Solutions Book.

Example:

Internal Taper: Given b = 1.150'' r_1 = .21875''
c = 1.050'' r_2 = .34375''
d = .800''

Keystrokes:	Display:
[shift] [fix] 4	(set display mode)
[XEQ] [ALPHA] SIZE [ALPHA] 005	
[XEQ] [ALPHA] TAPERS [ALPHA]	IN. ,EX. TAPERS
[A]	R1?
.21875 [R/S]	R2?
.34375 [R/S]	c?
1.05 [R/S]	d?
.8 [R/S]	b?
1.15 [R/S]	C=0.2806
[R/S]	D=0.5099
[R/S]	PHI=11.2753 (degs)
[R/S]	R1=0.4601

User Instructions

SIZE: 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set display mode			
2	Initialize the program.		[XEQ] TAPERS	IN. ,EX. TAPERS
3	Determine the case from the drawings.			
4	For internal taper:		[A]	R1?
		r_1	[R/S]	R2?
		r_2	[R/S]	c?
		c	[R/S]	d?
		d	[R/S]	b?
		b	[R/S]	C=(c)
			[R/S]*	D=(d)
			[R/S]*	PHI=(ϕ)
			[R/S]*	R1=(R1)
5	For an external taper, case 1:		[B]	R1?
		r_1	[R/S]	R2?
		r_2	[R/S]	a?
		a	[R/S]	b?
		b	[R/S]	X=(x)
			[R/S]*	PHI=(ϕ)
6	For an external taper, case 2:		[C]	H?
		h	[R/S]	b?
		b	[R/S]	a?
		a	[R/S]	R?
		r	[R/S]	X=(x)
			[R/S]*	PHI=(ϕ)
7	For an external taper, case 3:		[D]	R1?
		r_1	[R/S]	R2?
		r_2	[R/S]	c?

User Instructions

SIZE: 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
		c	[R/S]	d?
		d	[R/S]	X=(x)
			[R/S]*	PHI=(ϕ)

* [R/S] is omitted when printer is attached

Program Listings

```
 01◆LBL "TAP
ERS"
 02 CF 01
 03 SF 21
 04 SF 27
 05 DEG
 06 "IN.,EX.
 TAPERS"
 07 AVIEW
 08 ADV
 09 STOP
 10◆LBL A
 11 "R1?"
 12 PROMPT
 13 STO 00
 14 "R2?"
 15 PROMPT
 16 +
 17 ST+ X
 18 STO 01
 19 "c?"
 20 PROMPT
 21 *
 22 LASTX
 23 X↑2
 24 -
 25 SQRT
 26 STO 02
 27 RCL 01
 28 "d?"
 29 PROMPT
 30 *
 31 LASTX
 32 X↑2
 33 -
 34 SQRT
 35 STO 04
 36 RCL 02
 37 -
 38 "b?"
 39 PROMPT
 40 /
 41 ATAN
 42 STO 03
 43 90
 44 +
 45 2
 46 /
 47 TAN
 48 1/X
 49 ST* 00
```

Initialize

Internal tapers

Input r_1, r_2, c, d and b

Calculate C, D, ϕ, R1

```
 50 RCL 02
 51 "C"
 52 XEQ 11
 53 RCL 04
 54 "D"
 55 XEQ 11
 56 RCL 03
 57 "PHI"
 58 XEQ 11
 59 RCL 02
 60 RCL 00
 61 +
 62 "R1"
 63 GTO 11
 64◆LBL B
 65 "R1?"
 66 PROMPT
 67 STO 00
 68 "R2?"
 69 PROMPT
 70 STO 01
 71 -
 72 "a?"
 73 FS? 01
 74 "c?"
 75 PROMPT
 76 RCL 00
 77 -
 78 STO 02
 79 "b?"
 80 FS? 01
 81 "d?"
 82 PROMPT
 83 RCL 01
 84 -
 85 -
 86 /
 87 FS?C 01
 88 RTN
 89 ATAN
 90 STO 03
 91 LASTX
 92 1/X
 93 RCL 00
 94 *
 95 RCL 02
 96 -
 97 CHS
 98 90
 99 RCL 03
100 ST+ X
```

External taper case 1

Input r_1, r_2, a and b

Calculate x & ϕ

Program Listings

```
101 -
102 X<>Y
103 GTO 05
104◆LBL C
105 "H?"
106 PROMPT
107 "b?"
108 PROMPT
109 "a?"
110 PROMPT
111 STO 00
112 -
113 /
114 ATAN
115 STO 01
116 CHS
117 90
118 +
119 RCL 00
120 RCL 01
121 2
122 /
123 TAN
124 1/X
125 1
126 +
127 "R?"
128 PROMPT
129 *
130 -
131◆LBL 05
132 "X"
133 XEQ 11
134 RDN
135 "PHI"
136 GTO 11
137◆LBL D
138 SF 01
139 XEQ B
140 ASIN
141 STO 03
142 RCL 00
143 LASTX
144 RCL 02
145 *
146 -
147 RCL 03
148 COS
149 /
150 GTO 05
```

External taper case 2

Input h, b, a and r

Calculate x, ϕ

Display x, ϕ

External taper case 3

Input r_1, r_2, c and d

Calculate x, ϕ

```
151◆LBL 11
152 "⊢="
153 ARCL X
154 AVIEW
155 END
```

Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	r_1	50	
	$(r_1 + r_2) * 2$		
	C		Internal taper
	D		
	tan ϕ		
		55	
00	r_1		
	r_2		
	$a - r_1$		External taper, case 1
	θ		
		60	
00	a		
	2θ		External taper, case 2
00	r_1		
	r_2	65	
	$c - r_1$		External taper, case 3
	ϕ		
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 005 TOT. REG. 41 USER MODE

ENG ____ FIX ____ SCI ____ ON X OFF ____

DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01	C	Case 3	Case 1
21	S	Printer enable	Printer disable
27	S	User mode on	User mode off

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

INTERNAL AND EXTERNAL TAPERS

PROGRAM REGISTERS NEEDED: 37

HEWLETT PACKARD
SOLUTIONS BOOK:
GEOMETRY

ROW 1 (1 : 3)

ROW 2 (3 : 6)

ROW 3 (6 : 11)

ROW 4 (11 : 18)

ROW 5 (19 : 28)

ROW 6 (28 : 38)

ROW 7 (39 : 49)

ROW 8 (50 : 56)

ROW 9 (57 : 62)

ROW 10 (63 : 68)

ROW 11 (68 : 75)

ROW 12 (76 : 83)

ROW 13 (84 : 95)

ROW 14 (96 : 104)

ROW 15 (105 : 111)

ROW 16 (112 : 123)

ROW 17 (124 : 133)

ROW 18 (133 : 138)

ROW 19 (139 : 149)

ROW 20 (150 : 155)

POINTS OF TANGENCY WITH CIRCLES AND ARCS

This program will accurately locate points of tangency between straight lines and arcs, between straight lines and a circle, and between two circles and a straight line. All angular outputs are in decimal degrees.

Solutions for Finding Point of Tangency With an Arc:

Given: a and r, determine x and y

$$b^2 = (a + r)^2 - r^2$$

$$\sin\theta = \frac{r}{a + r} = \frac{y}{b} = \frac{x}{r}$$

$$x = \frac{r^2}{a + r}$$

$$y = \frac{br}{a + r}$$



Solution for Finding Points of Tangency with A Circle:

Given: b, c and r, determine x_1 and y_1

$$a = \sqrt{b^2 + c^2} - r$$

$$\sin\theta = \frac{r}{a + r}$$

$$\tan\alpha = \frac{b}{c}$$

$$\gamma = 90° - \theta - \alpha$$

$$e = r\sin\gamma, \text{ then } x_1 = b + e$$

$$d = r\cos\gamma, \text{ then } y_1 = c - d$$



Solution for Finding Points of Tangency with Two Circles:

Given: a, b, r_1 and r_2, determine x_1, y_1, x_2 and y_2

$$c = a^2 + b^2$$

$$\tan \theta = \frac{b}{a}$$

$$\sin \phi = \frac{r_2 - r_1}{c}$$

$$x_1 = r_1 \sin(\theta + \phi)$$

$$y_1 = r_1 \cos(\theta + \phi)$$

$$x_2 = r_2 \sin(\theta + \phi)$$

$$y_2 = b + r_2 \cos(\theta + \phi)$$



Reference: HP-67/97 "Geometry" Users' Library Solutions Book.

Example:

Find the point of tangency with an arc given: a = 1.125'' and r = .750''.

Keystrokes:	Display:
[XEQ] [ALPHA] DEG [ALPHA]	(set angular mode)
[XEQ] [ALPHA] SIZE [ALPHA] 007	
[XEQ] [ALPHA] POINTS [ALPHA]	POINTS OF T.
[A]	a?
1.125 [R/S]	R?
.75 [R/S]	X=0.3000
[R/S]	Y=0.6874
[R/S]	b=1.7185
[R/S]	THETA=23.5782 (degs)

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set angular mode			
2	Initialize the program.		[XEQ] POINTS	POINTS OF T.
3	Determine the case from the drawings.			
4	For the point of tangency with an arc:		[A]	a?
		a	[R/S]	R?
		r	[R/S]	X=(x)
			[R/S] *	Y=(y)
	(optional)		[R/S] *	b=(b)
			[R/S] *	THETA=(θ)
5	For the point of tangency with a circle:		[B]	b?
		b	[R/S]	c?
		c	[R/S]	R?
		r	[R/S]	X1=(x_1)
			[R/S] *	Y1=(y_1)
	(optional)		[R/S] *	a=(a)
			[R/S] *	THETA=(θ)
			[R/S] *	ALPHA=(α)
6	For the points of tangency with two			
	circles:		[C]	a?
		a	[R/S]	b?
		b	[R/S]	R2?
		r_2	[R/S]	R1?
		r_1	[R/S]	X1=(x_1)
			[R/S] *	Y1=(y_1)
			[R/S] *	X2=(x_2)
			[R/S] *	Y2=(y_2)
	(optional)		[R/S] *	c=(c)

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
			[R/S] *	THETA=(θ)
			[R/S] *	PHI=(ϕ)
	*[R/S] not required when printer attached			

Program Listings

```
 01◆LBL "POI
NTS"
 02 SF 21
 03 SF 27
 04 "POINTS
OF T."
 05 AVIEW
 06 STOP
 07◆LBL A
 08 "a?"
 09 PROMPT
 10 "R?"
 11 PROMPT
 12 STO 01
 13 +
 14 STO 00
 15 X↑2
 16 RCL 01
 17 X↑2
 18 -
 19 SQRT
 20 STO 02
 21 RCL 01
 22 RCL 00
 23 /
 24 *
 25 LASTX
 26 RCL 01
 27 *
 28 "X"
 29 XEQ 11
 30 RDN
 31 "Y"
 32 XEQ 11
 33 RCL 02
 34 "b"
 35 XEQ 11
 36 /
 37 ASIN
 38 "THETA"
 39 GTO 11
 40◆LBL B
 41 "b?"
 42 PROMPT
 43 STO 00
 44 "c?"
 45 PROMPT
 46 STO 01
 47 R-P
 48 "R?"
 49 PROMPT
```

Initialize

Tangency with an arc

Input a, r

Calculate x, y, b, θ

Tangency with a circle

Input b, c, r

```
 50 STO 04
 51 -
 52 STO 02
 53 LASTX
 54 LASTX
 55 RCL 02
 56 +
 57 /
 58 ASIN
 59 STO 03
 60 90
 61 -
 62 CHS
 63 RCL 00
 64 RCL 01
 65 /
 66 ATAN
 67 STO 05
 68 -
 69 RCL 04
 70 P-R
 71 RCL 01
 72 -
 73 CHS
 74 X<>Y
 75 RCL 00
 76 +
 77 "X1"
 78 XEQ 11
 79 RDN
 80 "Y1"
 81 XEQ 11
 82 RCL 02
 83 "a"
 84 XEQ 11
 85 RCL 03
 86 "THETA"
 87 XEQ 11
 88 RCL 05
 89 "ALPHA"
 90 GTO 11
 91◆LBL C
 92 "a?"
 93 PROMPT
 94 STO 00
 95 "b?"
 96 PROMPT
 97 STO 01
 98 R-P
 99 STO 02
100 "R2?"
```

Calculate x_1, y_1, a, θ, α

Tangency with two circles

Program Listings

```
101 PROMPT
102 STO 04
103 "R1?"
104 PROMPT
105 STO 05
106 -
107 /
108 1/X
109 ASIN
110 STO 03
111 RCL 01
112 RCL 00
113 /
114 ATAN
115 STO 00
116 +
117 STO 06
118 RCL 05
119 P-R
120 X<>Y
121 "X1"
122 XEQ 11
123 X<>Y
124 "Y1"
125 XEQ 11
126 RCL 06
127 RCL 04
128 P-R
129 RCL 01
130 +
131 X<>Y
132 "X2"
133 XEQ 11
134 X<>Y
135 "Y2"
136 XEQ 11
137 RCL 02
138 "c"
139 XEQ 11
140 RCL 00
141 "THETA"
142 XEQ 11
143 RCL 03
144 "PHI"
145◆LBL 11
146 "⊢="
147 ARCL X
148 AVIEW
149 END
```

Input a, b, r_1, r_2

Calculate x_1, y_1, x_2, y_2, c, θ, ϕ

Display routine

51		
60		
70		
80		
90		
00		

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	a + r	50	
	r		with an arc
	b		
00	b	55	
	c		
	a		
	θ		with a circle
	r		
05	α	60	
00	a or θ		
	b		
	c	65	
	ϕ		with two circles
	r_2		
05	r_1		
	θ + ϕ		
		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 45 USER MODE

ENG ____ FIX ____ SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21	S	Printer enable	Printer disable
27	S	User mode on	User mode off

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
An arc	A	Two circles	C
A circle	B		

POINTS OF TANGENCY WITH CIRCLES AND ARCS
PROGRAM REGISTERS NEEDED: 39

HEWLETT PACKARD
SOLUTIONS BOOK:
GEOMETRY

ROW 1 (1 : 3)

ROW 2 (3 : 4)

ROW 3 (4 : 11)

ROW 4 (12 : 24)

ROW 5 (25 : 32)

ROW 6 (32 : 38)

ROW 7 (38 : 44)

ROW 8 (44 : 54)

ROW 9 (55 : 66)

ROW 10 (67 : 77)

ROW 11 (78 : 83)

ROW 12 (84 : 87)

ROW 13 (88 : 92)

ROW 14 (92 : 100)

ROW 15 (100 : 109)

ROW 16 (110 : 121)

ROW 17 (121 : 127)

ROW 18 (128 : 135)

POINTS OF TANGENCY WITH CIRCLES AND ARCS

HEWLETT PACKARD SOLUTIONS BOOK: GEOMETRY

ROW 19 (135 : 141)

ROW 20 (141 : 145)

ROW 21 (146 : 149)

LINE-LINE INTERSECTION

This program will calculate the point of intersection of two lines. For each line the user specifies two points, or one point and the angle from horizontal, or one point and the slope. Slope will be converted to angle by the relation $\theta = \tan^{-1}(\text{slope})$. Given two points (x_1,y_1) and (x_2,y_2) on the line, the angle is:

$$\theta = \tan^{-1} \frac{y_2 - y_1}{x_2 - x_1}$$

(x,y) = Coordinates of point of intersection

(x_1,y_1) = Coordinates of point on line one

(x_2,y_2) = Coordinates of point on line two

θ_1 = Angle from horizontal to line one

θ_2 = Angle from horizontal to line two



Equations:

$$x = \frac{x_1 \tan\theta_1 - x_2 \tan\theta_2 + y_2 - y_1}{\tan\theta_1 - \tan\theta_2}$$

$$y = y_1 + (x - x_1)\tan\theta_1$$

Reference: HP-67/97 "Geometry" Users' Library Solutions Book.

Example:

Find the point of intersection of two lines, one passing through (10,20) (40,30), and the other through (-10,30) (50,10).

Keystrokes:	Display:
[shift] [fix] 4	(set display mode)
[XEQ] [ALPHA] SIZE [ALPHA] 007	
[XEQ] [ALPHA] LINE [ALPHA]	LINE INTRSEC
[A]	X1 ?
10 [R/S]	Y1 ?
20 [R/S]	X2 ?
40 [R/S]	Y2 ?
30 [R/S]	NEXT LINE ?
[A]	X1 ?
10 [CHS] [R/S]	Y1 ?
30 [R/S]	X2 ?
50 [R/S]	Y2 ?
10 [R/S]	X=15.0000
[R/S]	Y=21.6667

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set display mode			
2	Initialize the program.		[XEQ] LINE	LINE INTRSEC
	LINE-LINE INTERSECTION:			
3a	Input two points on line:		[A]	X1 ?
		x_1	[R/S]	Y1 ?
		y_1	[R/S]	X2 ?
		x_2	[R/S]	Y2 ?
		y_2	[R/S]	NEXT LINE ?
3b	Or, input one point and the slope, m:		[B]	X ?
		x	[R/S]	Y ?
		y	[R/S]	M ?
		m	[R/S]	NEXT LINE ?
3c	Or, input one point and the angle θ:		[C]	X ?
		x	[R/S]	Y ?
		y	[R/S]	THETA ?
		θ	[R/S]	NEXT LINE ?
3d	Or, for the case where the second line is			
	vertical, input the x coordinate:	x	[D]	Y=(y)
4	Repeat step 3 for the second line.			
5	After the parameters for the second line			
	are input, the intersection coordinates			X=(x)
	are automatically displayed.		[R/S]	Y=(y)
6	For a new case, go to step 2.			

Program Listings

```
 01◆LBL "LIN
E"
 02 CF 00
 03 SF 21
 04 SF 27
 05 DEG
 06 1.006
 07 STO 00
 08 "LINE IN
TRSEC"
 09◆LBL 01
 10 AVIEW
 11 STOP
 12◆LBL A
 13 "X1 ?"
 14 PROMPT
 15 STO IND
00
 16 ISG 00
 17 "Y1 ?"
 18 PROMPT
 19 STO IND
00
 20 X<>Y
 21 "X2 ?"
 22 PROMPT
 23 -
 24 X<>Y
 25 "Y2 ?"
 26 PROMPT
 27 -
 28 /
 29 1/X
 30 GTO 02
 31◆LBL C
 32 SF 00
 33◆LBL B
 34 "X ?"
 35 PROMPT
 36 STO IND
00
 37 ISG 00
 38 "Y ?"
 39 PROMPT
 40 STO IND
00
 41 "M ?"
 42 FS? 00
 43 "THETA ?
"
 44 PROMPT
```

Initialize flags

Store loop control value

Two points on line input routine

Input x_1, y_1 and x_2, y_2

Calculate θ

Input routine for x, y and slope or θ

```
 45 FS?C 00
 46 TAN
 47◆LBL 02
 48 ISG 00
 49 STO IND
00
 50 "NEXT LI
NE ?"
 51 ISG 00
 52 GTO 01
 53 RCL 01
 54 RCL 03
 55 *
 56 RCL 04
 57 RCL 06
 58 *
 59 -
 60 RCL 05
 61 +
 62 RCL 02
 63 -
 64 RCL 03
 65 RCL 06
 66 -
 67 /
 68 "X"
 69 XEQ 11
 70◆LBL D
 71 RCL 01
 72 -
 73 RCL 03
 74 *
 75 RCL 02
 76 +
 77 "Y"
 78◆LBL 11
 79 "⊢="
 80 ARCL X
 81 AVIEW
 82 END
```

Get input for second line

Calculate x and y (intersect point)

Calculate x

Calculate y

Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	loop control	50	
	x_1 - line 1		
	y_1 - line 1		
	tan θ_1 or m_1		
	x_2 - line 2		
05	y_1 - line 2	55	
	tan θ_2 or m_2		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 32 USER MODE

ENG ____ FIX ____ SCI ____ ON X OFF ____

DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00	C	Input slope	Input theta
21	S	Printer enable	Printer disable
27	S	User mode on	User mode off

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
Input 2 points	A	Input 1 pt & θ	C
Input 1 pt & m	B		

LINE–LINE INTERSECTION

PROGRAM REGISTERS NEEDED: 26

HEWLETT PACKARD
SOLUTIONS BOOK:
GEOMETRY

ROW 1 (1 : 4)

ROW 2 (4 : 8)

ROW 3 (8 : 12)

ROW 4 (13 : 17)

ROW 5 (17 : 23)

ROW 6 (24 : 31)

ROW 7 (31 : 37)

ROW 8 (37 : 42)

ROW 9 (42 : 46)

ROW 10 (47 : 50)

ROW 11 (50 : 57)

ROW 12 (58 : 69)

ROW 13 (69 : 78)

ROW 14 (79 : 82)

POINTS ON A STRAIGHT LINE

This program calculates the coordinates of equidistant points on a straight line.

Equations:

Point P_i is calculated by

$$x_i = x_1 + (i - 1)\,H \cos\theta$$

$$y_i = y_1 + (i - 1)\,H \sin\theta, \quad i = \pm 0, 1, 2, \ldots$$

where

$P_1 = (x_1, y_1)$ (the starting point);

θ is the angle of the straight line with the x axis;

H is the distance between consecutive points in the direction of the straight line.



Reference: HP-67/97 "Geometry" Users' Library Solutions Book.

Example:

For the straight line designated by $X_1 = 10$, $Y_1 = 10$, $\theta = -30°$, calculate P_i for H = 20 and i = 1, 2, and 3.

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 005	
[XEQ] [ALPHA] PLINE [ALPHA]	PTS. ON ST. L.
	X1 ?
10 [R/S]	Y1 ?
10 [R/S]	THETA ?
30 [CHS] [R/S]	H ?
20 [R/S]	I ?
1 [R/S]	X=10.0000
[R/S]	Y=10.0000
[R/S] [R/S]	X=27.3205
[R/S]	Y=0.0000
[R/S] [R/S]	X=44.6410
[R/S]	Y=-10.0000

User Instructions

SIZE: 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] PLINE	PTS. ON ST. L.
				X1 ?
3	Input P_1, the angle of the line, and the	x_1	[R/S]	Y1 ?
	distance between points.	y_1	[R/S]	THETA ?
		θ	[R/S]	H ?
		H	[R/S]	I ?
4	Calculate P_i.	i	[R/S]	X=(x)
			[R/S]	Y=(y)
5a	To calculate a new P_i,		[R/S]	I ?
	and go to step 4, or			
5b	To calculate $P_{(i+1)}$		[R/S]	I ?
			[R/S]	X=(x)
			[R/S]	Y=(y)

Program Listings

```
 01*LBL "PLI
NE"
 02 "PTS. ON
 ST. L."
 03 AVIEW
 04 PSE
 05 "X1 ?"
 06 PROMPT
 07 STO 02
 08 "Y1 ?"
 09 PROMPT
 10 STO 03
 11 "THETA ?
"
 12 PROMPT
 13 "H ?"
 14 PROMPT
 15 P-R
 16 STO 00
 17 RDN
 18 STO 01
 19*LBL 01
 20 "I ?"
 21 PROMPT
 22 1
 23 -
 24 STO 04
 25 RCL 00
 26 *
 27 RCL 02
 28 +
 29 "X"
 30 XEQ 11
 31 RCL 04
 32 RCL 01
 33 *
 34 RCL 03
 35 +
 36 "Y"
 37 XEQ 11
 38 RCL 04
 39 2
 40 +
 41 GTO 01
 42*LBL 11
 43 "⊢="
 44 ARCL X
 45 AVIEW
 46 STOP
 47 RTN
 48 .END.
```

Initialize

Input x_1, y_1, θ, and H

Input I

Calculate x, y

Display routine

51 60 70 80 90 00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	h cos	50	
	h sin		
	x_1		
	y_1		
	i+1		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 005 TOT. REG. 19 USER MODE

ENG ____ FIX ____ SCI ____ ON X OFF ____

DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

POINTS ON A STRAIGHT LINE

PROGRAM REGISTERS NEEDED: 15

ROW 1 (1 - 2)

ROW 2 (2 - 4)

ROW 3 (5 - 9)

ROW 4 (10 - 13)

ROW 5 (13 - 22)

ROW 6 (23 - 32)

ROW 7 (33 - 41)

ROW 8 (42 - 48)

GRID OF POINTS: CALCULATE ALL POINTS

This program calculates the X and Y coordinates of all the points on a grid defined as follows:

a. First direction of a grid:

the angle, θ_1, with the positive X axis
the algebraic distance between each point, H_1, in this direction
the total number, N_1, of points (including the first one)

b. Second direction of the grid:

the angle, θ_2, with the positive X axis
the algebraic distance between two points, H_2, in that direction
the total number, N_2, of points (including the first one)

c. Starting point (noted 1) with coordinates X and Y.

The calculation is incremental from point 1 to point (N_1,N_2). For each point we find:

The index i, the X_i and Y_i coordinates



Reference: HP-67/97 "Geometry" Users' Library Solutions Book.

Example:

Find the grid points for:

$\theta_1 = 0°$, $H_1 = 10$, $N_1 = 3$, $X_1 = 10$

$\theta_2 = 90°$, $H_2 = 20$, $N_2 = 2$, $Y_2 = 10$

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] GRIDALL [ALPHA]	GRID ALL PTS
	N1 ?
3 [R/S]	N2 ?
2 [R/S]	X1 ?
10 [R/S]	Y1 ?
10 [R/S]	H1 ?
10 [R/S]	H2 ?
20 [R/S]	THETA 1 ?
0 [R/S]	THETA 2 ?
90 [R/S]	X1=10.0000
[R/S]	Y1=10.0000
[R/S]	X2=20.0000
[R/S]	Y2=10.0000
⋮	⋮
[R/S]	X6=30.0000
[R/S]	Y6=30.0000
[R/S]	END

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] GRIDALL	GRID ALL PTS
				N1 ?
3	Input the number of points:	N1	[R/S]	N2 ?
		N2	[R/S]	X1 ?
	the origin:	X1	[R/S]	Y1 ?
		Y1	[R/S]	H1 ?
	grid increments:	H1	[R/S]	H2 ?
		H2	[R/S]	THETA 1 ?
	and directions.	θ1	[R/S]	THETA 2 ?
	Calculate the coordinates of the grid	θ2	[R/S]	$X1=(x_1)$
	points.		[R/S]	$Y1=(y_1)$
			[R/S]	$X(i)=(x_i)$
			[R/S]	$Y(i)=(y_i)$
			⋮	⋮
			[R/S]	END
4	For another case, go to step 2.			

Program Listings

```
 01*LBL "GRI
DALL"
 02 1
 03 STO 09
 04 CF 29
 05 "GRID AL
L PTS"
 06 AVIEW
 07 PSE
 08 "N1 ?"
 09 PROMPT
 10 1
 11 -
 12 1 E3
 13 /
 14 STO 06
 15 STO 08
 16 "N2 ?"
 17 PROMPT
 18 1
 19 -
 20 1 E3
 21 /
 22 STO 07
 23 "X1 ?"
 24 PROMPT
 25 STO 00
 26 "Y1 ?"
 27 PROMPT
 28 STO 01
 29 "H1 ?"
 30 PROMPT
 31 +
 32 "H2 ?"
 33 PROMPT
 34 STO 04
 35 "THETA 1
?"
 36 PROMPT
 37 LASTX
 38 P-R
 39 STO 02
 40 RDN
 41 STO 03
 42 "THETA 2
?"
 43 PROMPT
 44 RCL 04
 45 P-R
 46 STO 04
 47 RDN
```

Initialize

Input N1, N2, X1, X2, H1, H2 θ1, θ2

Calculate ΔX's and ΔY's

```
 48 STO 05
 49 GTO d
50*LBL 01
 51 1
 52 ST+ 09
 53 ISG 06
 54 GTO d
 55 RCL 08
 56 STO 06
 57 ISG 07
 58 GTO d
 59 "END"
 60 AVIEW
 61 STOP
62*LBL d
 63 RCL 06
 64 INT
 65 RCL 02
 66 *
 67 RCL 04
 68 RCL 07
 69 INT
 70 *
 71 +
 72 RCL 00
 73 +
 74 "X"
 75 XEQ 12
 76 RCL 03
 77 RCL 06
 78 INT
 79 *
 80 RCL 05
 81 RCL 07
 82 INT
 83 *
 84 +
 85 RCL 01
 86 +
 87 "Y"
 88 XEQ 12
 89 GTO 01
90*LBL 12
 91 FIX 0
 92 ARCL 09
 93 "⊢="
 94 FIX 4
 95 ARCL X
 96 AVIEW
 97 STOP
 98 RTN
```

Loop control routine

Calculate X, Y

Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	X1	50	
	Y1		
	ΔX_1		
	ΔY_1		
	ΔX_2		
05	ΔY_2	55	
	$1.00N_1-1$		
	$1.00N_2-1$		
	$.00N_1-1$		
	Counter		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 37 USER MODE
ENG ___ FIX ___ SCI ___ ON X OFF ___
DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
29	C	For proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

GRID OF POINTS:
CALCULATE ALL POINTS
PROGRAM REGISTERS NEEDED: 28

HEWLETT PACKARD
SOLUTION BOOK:
GEOMETRY

ROW 1 (1 - 3)

ROW 2 (4 - 5)

ROW 3 (5 - 12)

ROW 4 (12 - 19)

ROW 5 (20 - 26)

ROW 6 (26 - 31)

ROW 7 (32 - 35)

ROW 8 (35 - 42)

ROW 9 (42 - 48)

ROW 10 (49 - 55)

ROW 11 (56 - 62)

ROW 12 (62 - 74)

ROW 13 (74 - 84)

ROW 14 (85 - 92)

ROW 15 (92 - 99)

ROW 16 (99 - 99)

GRID OF POINTS: CALCULATE DISCRETE POINTS

This program calculates the cartesian coordinates of specified points of a grid defined as follows:

a. First direction:

the angle θ_1 (related to positive X axis)

the distance between each point, H_1, in this direction

b. Second direction:

the angle θ_2

the distance H_2

c. Starting point (origin of the grid), X_{11} and Y_{11} .

Formulas:

$$Xij = X_1 + (j-1)\ \Delta X_1 + (i-1)\ \Delta X_2$$

$$Yij = Y_1 + (j-1)\ \Delta Y_1 + (i-1)\ \Delta Y_2$$

where $\Delta X_1 = H_1 \cos \theta_1$

$\Delta Y_1 = H_1 \sin \theta_1$

$\Delta X_2 = H_2 \cos \theta_2$

$\Delta Y_2 = H_2 \sin \theta_2$



Reference: HP-67/97 "Geometry" Users' Library Solutions Book.

Example:

For a grid with its origin at (1,1), $H_1 = 2$, $H_2 = 3$, $\theta_1 = 30$, and $\theta_2 = 90°$, find the cartesian coordinates for the following grid coordinates: (1,1), (2.5,4).

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 008	
[XEQ] [ALPHA] GRIDISC [ALPHA]	GRID DIS. PTS.
	X1 ?
1 [R/S]	Y1 ?
1 [R/S]	H1 ?
2 [R/S]	H2 ?
3 [R/S]	THETA 1 ?
30 [R/S]	THETA 2 ?
90 [R/S]	I?
1 [R/S]	J?
1 [R/S]	X=1.0000
[R/S]	Y=1.0000
[A]	I?
2.5 [R/S]	J?
4 [R/S]	X=6.1962
[R/S]	Y=8.5000

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] GRIDISC	GRID DIS. PTS.
3	Input grid parameters:			X1 ?
		x_1	[R/S]	Y1 ?
		y_1	[R/S]	H1 ?
		H_1	[R/S]	H2 ?
		H_2	[R/S]	THETA 1 ?
		θ_1	[R/S]	THETA 2 ?
		θ_2	[R/S]	I?
4	Input i, j, and calculate grid x,y			
	coordinates.	i	[R/S]	J?
		j	[R/S]	X=(x)
			[R/S]	Y=(y)
5	For additional points on the grid:		[A]	I?
	then go to step 4.			
6	For a new case, go to step 2.			

Program Listings

```
 01*LBL "GRI
DISC"
 02 "GRID DI
S. PTS."
 03 AVIEW
 04 PSE
 05 "X1 ?"
 06 PROMPT
 07 STO 00
 08 "Y1 ?"
 09 PROMPT
 10 STO 01
 11 "H1 ?"
 12 PROMPT
 13 +
 14 "H2 ?"
 15 PROMPT
 16 STO 04
 17 "THETA 1
?"
 18 PROMPT
 19 LASTX
 20 P-R
 21 STO 02
 22 RDN
 23 STO 03
 24 "THETA 2
?"
 25 PROMPT
 26 RCL 04
 27 P-R
 28 STO 04
 29 RDN
 30 STO 05
 31*LBL A
 32 "I?"
 33 PROMPT
 34 1
 35 -
 36 STO 06
 37 "J?"
 38 PROMPT
 39 1
 40 -
 41 STO 07
 42 RCL 02
 43 *
 44 RCL 04
 45 RCL 06
 46 *
 47 +
```

Initialize

Input x_1, y_2, H1, H2, , and calculate θx's and θy's

Input i, j and calculate x, y

```
 48 RCL 00
 49 +
 50 "X"
 51 XEQ 11
 52 RCL 03
 53 RCL 07
 54 *
 55 RCL 05
 56 RCL 06
 57 *
 58 +
 59 RCL 01
 60 +
 61 "Y"
 62*LBL 11
 63 "⊢="
 64 ARCL X
 65 AVIEW
 66 STOP
 67 RTN
 68 .END.
```

Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	X_1	50	
	Y_1		
	ΔX_1		
	ΔY_1		
	ΔX_2		
05	ΔY_2	55	
	I-1		
	J-1		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 27 USER MODE

ENG ____ FIX ____ SCI ____ ON X OFF ____

DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

GRID OF POINTS:
CALCULATE DISCRETE POINTS
PROGRAM REGISTERS NEEDED: 20

HEWLETT PACKARD
SOLUTION BOOK:
GEOMETRY

ROW 1 (1 - 2)

ROW 2 (2 - 2)

ROW 3 (3 - 8)

ROW 4 (8 - 14)

ROW 5 (14 - 17)

ROW 6 (17 - 24)

ROW 7 (24 - 32)

ROW 8 (32 - 41)

ROW 9 (42 - 51)

ROW 10 (52 - 63)

ROW 11 (63 - 68)

TANGENT CIRCLE TO TWO STRAIGHT LINES WITH A GIVEN RADIUS

This program calculates the X and Y coordinates of the centers of the four circles with a given radius, R, which are tangent to two given lines.



The straight lines are each defined by one point and an angle which follows the convention below:



The straight lines are first shifted by R. The calculation is then one of the intersection of two straight lines.

Formulas used:

$$X = \frac{(Y_2 - Y_1)\cos\theta_1 \cos\theta_2 + X_1 \sin\theta_1 \cos\theta_2 - X_2 \sin\theta_2 \cos\theta_1}{\sin(\theta_1 - \theta_2)}$$

$$Y = Y_1 + (X - X_1)\tan\theta_1, \quad |\theta| \geq 90°$$

$$Y = Y_2 + (X - X_2)\tan\theta_2, \quad |\theta| < 90°$$

Reference: HP-67/97 "Geometry" Users' Library Solutions Book.

Example:

Find the tangent circle for:

$D_1 = [10,20,30°]$ $D_2 = [-20,30,-60°]$ $R = 10$

Executing the program four times will yield:

	X	Y
Above (A)	-4.5096	23.1699
Below (B)	-11.8301	-4.1506
Left (L)	-21.8301	13.1699
Right (R)	5.4904	5.8494

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] TANGENT [ALPHA]	TANGENT CIRC.
	R?
10 [R/S]	WHERE(L,R,A,B) ?
A [R/S]	X?
10 [R/S]	Y?
20 [R/S]	THETA?
30 [R/S]	X?
20 [CHS] [R/S]	Y?
30 [R/S]	THETA ?
60 [CHS] [R/S]	X=-4.5096
[R/S]	Y=23.1699

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program.			
2	Initialize the program.		[XEQ] TANGENT	TANGENT CIRC.
				R?
3	Input R.	R	[R/S]	WHERE (L,R,A,B)?
4	Input position L = left; R = right;			
	A = above; B = below.	pos	[R/S]	X?
5	Input X_i, Y_i, θ_i.	X_1	[R/S]	Y?
		Y_1	[R/S]	THETA ?
		θ_1	[R/S]	X?
		X_2	[R/S]	Y?
		Y_2	[R/S]	THETA ?
		θ_2	[R/S]	X=(X)
			[R/S]	Y=(Y)
6	For another case, go to step 2.			

Program Listings

```
 01*LBL "TAN
GENT"
 02 CF 01
 03 CF 02
 04 CF 03
 05 CF 04
 06 "TANGENT
 CIRC."
 07 AVIEW
 08 PSE
 09 "R?"
 10 PROMPT
 11 STO 08
 12 "WHERE<L
,R,A,B>?"
 13 AON
 14 PROMPT
 15 AOFF
 16 ASTO Y
 17 "A"
 18 ASTO X
 19 X=Y?
 20 SF 02
 21 "B"
 22 ASTO X
 23 X=Y?
 24 SF 03
 25 "L"
 26 ASTO X
 27 X=Y?
 28 SF 04
 29*LBL 07
 30 "X?"
 31 PROMPT
 32 "Y?"
 33 PROMPT
 34 "THETA ?
"
 35 PROMPT
 36 FS? 02
 37 GTO B
 38 FS? 03
 39 GTO C
 40 FS? 04
 41 GTO D
 42 X<0?
 43 GTO B
 44 GTO C
 45*LBL D
 46 X<0?
 47 GTO C
```

Initialize

Input R

Input position

Input X, Y, θ

Set up calculations

```
 48*LBL B
 49 RCL 08
 50 GTO 01
 51*LBL C
 52 RCL 08
 53 CHS
 54*LBL 01
 55 X<>Y
 56 FS?C 01
 57 GTO 02
 58 STO 03
 59 X<>Y
 60 P-R
 61 X<>Y
 62 RDN
 63 +
 64 STO 02
 65 RDN
 66 X<>Y
 67 -
 68 STO 01
 69 SF 01
 70 GTO 07
 71*LBL 02
 72 STO 06
 73 X<>Y
 74 P-R
 75 X<>Y
 76 RDN
 77 +
 78 STO 05
 79 RDN
 80 X<>Y
 81 -
 82 STO 04
 83 RCL 05
 84 RCL 02
 85 -
 86 RCL 03
 87 COS
 88 STO 07
 89 *
 90 RCL 06
 91 COS
 92 STO 08
 93 *
 94 RCL 01
 95 RCL 08
 96 *
 97 RCL 03
 98 SIN
```

Calculate X_1, Y_1

Calculate X, Y

Calculate X

Program Listings

```
 99 *
100 +
101 RCL 04
102 RCL 07
103 *
104 RCL 06
105 SIN
106 *
107 -
108 RCL 03
109 RCL 06
110 -
111 SIN
112 /
113 STO 07
114 "X"
115 XEQ 11
116 RCL 03
117 ABS
118 90
119 X>Y?
120 GTO 03
121 RCL 07
122 RCL 01
123 -
124 RCL 03
125 TAN
126 *
127 RCL 02
128 GTO 04
129◆LBL 03
130 RCL 07
131 RCL 04
132 -
133 RCL 06
134 TAN
135 *
136 RCL 05
137◆LBL 04
138 +
139 "Y"
140◆LBL 11
141 "⊢="
142 ARCL X
143 AVIEW
144 STOP
145 RTN
146 .END.
```

Calculate Y

Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	X'_1		
	Y'_1		
	θ_1		
	X'_2		
05	Y'_2	55	
	θ_2		
	$\cos\theta_1$, X		
	R, $\cos\theta_2$		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 44 USER MODE
ENG ___ FIX ___ SCI ___ ON ___ OFF X
DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		1st pass, input X_1	2nd pass, input X_2, ...
02		circle above	
03		circle below	
04		circle left	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

TANGENT CIRCLE TO TWO STRAIGHT LINES WITH A GIVEN RADIUS
PROGRAM REGISTERS NEEDED: 36

ROW 1 (1 - 2)

ROW 2 (3 - 6)

ROW 3 (6 - 10)

ROW 4 (11 - 12)

ROW 5 (12 - 18)

ROW 6 (19 - 26)

ROW 7 (26 - 33)

ROW 8 (34 - 37)

ROW 9 (37 - 43)

ROW 10 (43 - 48)

ROW 11 (49 - 57)

ROW 12 (58 - 69)

ROW 13 (70 - 81)

ROW 14 (82 - 94)

ROW 15 (95 - 107)

ROW 16 (108 - 117)

ROW 17 (118 - 128)

ROW 18 (128 - 139)

TANGENT CIRCLE TO TWO STRAIGHT LINES WITH A GIVEN RADIUS

HEWLETT PACKARD SOLUTION BOOK: GEOMETRY

ROW 19 (140 - 146)

DISTANCE BETWEEN LINES IN SPACE

Given two lines, each defined by two points, this program calculates the shortest distance between the two lines. (This program was originally written to determine the clearance between electrical distribution circuits and guy wires or supporting structures).

The program takes lines defined by the two-point form,

$$\frac{x - x_1}{x_1' - x_1} = \frac{y - y_1}{y_1' - y_1} = \frac{z - z_1}{z_1' - z_1}$$

changes them to the point-direction form,

$$\frac{x - x_1}{a} = \frac{y - y_1}{b} = \frac{z - z_1}{c}$$

and the shortest distance (D) is calculated by:

$$D = \pm \frac{\begin{vmatrix} x_2 - x_1 & y_2 - y_1 & z_2 - z_1 \\ a_1 & b_1 & c_1 \\ a_2 & b_2 & c_2 \end{vmatrix}}{\sqrt{\begin{vmatrix} b_1 & c_1 \\ b_2 & c_2 \end{vmatrix}^2 + \begin{vmatrix} c_1 & a_1 \\ c_2 & a_2 \end{vmatrix}^2 + \begin{vmatrix} a_1 & b_1 \\ a_2 & b_2 \end{vmatrix}^2}}$$

Reference: Handbook of Tables for Mathematics, Third Edition, Samuel M. Selby, Published by The Chemical Rubber Co., 1967, page 509.

Example:

Given two lines in three-dimensional space:

Line #1 defined by points $(X_1,Y_1,Z_1) = (30,14,10)$ and $(X'_1,Y'_1,Z'_1) = (0,46,10)$;
Line #2 defined by points $(X_2,Y_2,Z_2) = (124,50,-30)$ and $(X'_2,Y'_2,Z'_2) = (0,36,16)$.

Calculate the shortest distance between the two lines.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 014	
[XEQ] [ALPHA] DIST [ALPHA]	DIST. B. LINES
	X1 ?
30 [R/S]	Y1 ?
14 [R/S]	Z1 ?
10 [R/S]	X1-PRIME ?
0 [R/S]	Y1-PRIME ?
46 [R/S]	Z1-PRIME ?
10 [R/S]	X2 ?
124 [R/S]	Y2 ?
50 [R/S]	Z2 ?
30 [CHS] [R/S]	X2-PRIME ?
0 [R/S]	Y2-PRIME ?
36 [R/S]	Z2-PRIME ?
16 [R/S]	D=2.5940

User Instructions

SIZE: 014

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program.			
2	Initialize the program.		[XEQ] DIST	DIST. B.LINES
3	Input line parameters:			X1 ?
		x_1	[R/S]	Y1 ?
		y_1	[R/S]	Z1 ?
		z_1	[R/S]	X1-PRIME ?
		x'_1	[R/S]	Y1-PRIME ?
		y'_1	[R/S]	Z1-PRIME ?
		z'_1	[R/S]	X2 ?
		x_2	[R/S]	Y2 ?
		y_2	[R/S]	Z2 ?
		z_2	[R/S]	X2-PRIME ?
		x'_2	[R/S]	Y2-PRIME ?
		y'_2	[R/S]	Z2-PRIME ?
	and calculate D.	z'_2	[R/S]	D=(D)*
4	For a new case, go to step 2.			
	*Negative sign, if present, has no meaning.			

Program Listings

```
 01*LBL "DIS
T"
 02 "DIST. B
. LINES"
 03 AVIEW
 04 2.2
 05 STO 00
 06 1.002
 07 STO 01
 08 CF 29
 09 FIX 0
 10*LBL 02
 11 "X"
 12 XEQ 12
 13 "Y"
 14 XEQ 12
 15 "Z"
 16 XEQ 12
 17 "X"
 18 XEQ 13
 19 "Y"
 20 XEQ 13
 21 "Z"
 22 XEQ 13
 23 ISG 01
 24 GTO 02
 25 RCL 08
 26 ST- 11
 27 RCL 09
 28 ST- 12
 29 RCL 10
 30 ST- 13
 31 RCL 02
 32 ST- 05
 33 ST- 08
 34 RCL 03
 35 ST- 06
 36 ST- 09
 37 RCL 04
 38 ST- 07
 39 ST- 10
 40 RCL 11
 41 RCL 06
 42 RCL 05
 43 RCL 12
 44 XEQ 14
 45 RCL 12
 46 RCL 07
 47 RCL 06
 48 RCL 13
 49 XEQ 14
```

Initialize

Input data

Calculate a_i, b_i, c_i, ΔX, ΔY, ΔZ

Calculate (A-B), (B-C) (C-A)

```
 50 RCL 13
 51 RCL 05
 52 RCL 07
 53 RCL 11
 54 XEQ 14
 55 RCL 05
 56 X↑2
 57 RCL 03
 58 X↑2
 59 +
 60 RCL 04
 61 X↑2
 62 +
 63 SQRT
 64 1/X
 65 RCL 08
 66 RCL 04
 67 *
 68 RCL 09
 69 RCL 05
 70 *
 71 +
 72 RCL 10
 73 RCL 03
 74 *
 75 +
 76 *
 77 FIX 4
 78 "D="
 79 ARCL X
 80 AVIEW
 81 STOP
 82*LBL 13
 83 ARCL 01
 84 "⊢-PRIME
 ?"
 85 GTO 15
 86*LBL 12
 87 ARCL 01
 88 "⊢ ?"
 89*LBL 15
 90 PROMPT
 91 STO IND
00
 92 ISG 00
 93 RTN
 94*LBL 14
 95 *
 96 STO IND
01
 97 RDN
```

Calculate D

Input prompting routine

Calculate 2x2 matrix

Program Listings

```
 98 *
 99 ST- IND
01
100 ISG 01
101 RTN
102 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	pointer	50	
	counter		
	x_1		
	y_1, (A-B)		
	z_1, (B-C)		
05	x'_1, a_1, (C-A)	55	
	y'_1, b_1		
	z'_1, c_1		
	x_2, x_2-x_1		
	y_2, y_2-y_1		
10	z_2, z_2-z_1	60	
	x'_2, a_2		
	y'_2, b_2		
	z'_2, c_2		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 014 TOT. REG. 41 USER MODE

ENG ___ FIX ___ SCI ___ ON ___ OFF X

DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
29	C	For proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

DISTANCE BETWEEN
LINES IN SPACE
PROGRAM REGISTERS NEEDED: 29

HEWLETT PACKARD
SOLUTION BOOK:
GEOMETRY

ROW 1 (1 - 2)

ROW 2 (2 - 4)

ROW 3 (4 - 10)

ROW 4 (11 - 16)

ROW 5 (16 - 21)

ROW 6 (21 - 28)

ROW 7 (28 - 36)

ROW 8 (36 - 44)

ROW 9 (45 - 54)

ROW 10 (54 - 66)

ROW 11 (67 - 78)

ROW 12 (78 - 84)

ROW 13 (84 - 88)

ROW 14 (88 - 95)

ROW 15 (96 - 102)

ROW 16 (102 - 102)



Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41 programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41, HP has an extensive library of "Application Pacs". These programs transform your HP-41 into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, miniature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41.

You can choose from:

Aviation (Pre-Flight Only) 00041-15018
Clinical Lab 00041-15024
Circuit Analysis 00041-15024
Financial Decisions 00041-15004
Mathematics 00041-15003
Structural Analysis 00041-15021
Surveying 00041-15005
Securities 00041-15026
Statistics 00041-15002
Stress Analysis 00041-15027
Games 00041-15022
Home Management 00041-15023
Machine Design 00041-15020
Navigation 00041-15017
Real Estate 00041-15016
Thermal and Transport Science 00041-15019
Petroleum Fluids 00041-15039

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41 Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

*Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales 00041-90094
Home Construction Estimating 00041-90096
Lending, Saving and Leasing 00041-90086
Real Estate 00041-90136
Small Business 00041-90137
Geometry 00041-90084
High-Level Math 00041-90083
Test Statistics 00041-90082
Antennas 00041-90093
Chemical Engineering 00041-90100
Control Systems 00041-90092
Electrical Engineering 00041-90088
Fluid Dynamics and Hydraulics 00041-90139
Games II 00041-90443
Civil Engineering 00041-90089
Heating, Ventilating & Air Conditioning 00041-90140
Mechanical Engineering 00041-90090
Solar Engineering 00041-90138
Calendars 00041-90145
Cardiac/Pulmonary 00041-90097
Chemistry 00041-90102
Games 00041-90099
Optometry I (General) 00041-90143
Optometry II (Contact Lens) 00041-90144
Physics 00041-90142
Surveying 00041-90141
Time Module Solutions 00041-90395

*Some books require additional memory modules to accomodate all programs.

GEOMETRY

SINE PLATE SOLUTIONS
V NOTCHES AND LONG RADII
INTERNAL AND EXTERNAL TAPERS
POINTS OF TANGENCY WITH CIRCLES AND ARCS
LINE-LINE INTERSECTION
POINTS ON A STRAIGHT LINE
GRID OF POINTS: CALCULATE ALL POINTS
GRID OF POINTS: CALCULATE DISCRETE POINTS
TANGENT CIRCLE TO TWO STRAIGHT LINES WITH A GIVEN RADIUS
DISTANCE BETWEEN LINES IN SPACE

00041-90084 Rev. B 6/83 Printed in U.S.A.